
	Reason for change:
	In the current TTCN implementation of fl_TC_8_2_3_16_1_ENDC_NR_Testbody it is observed that the TTCN configures the SRB3 with SDAP_Configuration_None.

However, as per 38.523-3, SDAP configuration Type is set to OMIT for ENDC testcases.

This needs to be corrected in TTCN implementation of above mentioned function as 8.2.3.16.1 is an ENDC testcase.

Similar issue exists for ENDC testcase 8.2.2.1.1 as well. This also needs to be corrected. 

	
	

	Summary of change:
	SDAP Configuration is set as Omit instead of None in the functions fl_TC_8_2_3_16_1_ENDC_NR_Testbody and fl_TC_8_2_2_1_1_ENDC_NR_Testbody. 

Please see below.

1.1
 Change 1 : fl_TC_8_2_3_16_1_ENDC_NR_TestBody
	Function name
	fl_TC_8_2_3_16_1_ENDC_NR_TestBody

	Reason for change
	In the current TTCN implementation of fl_TC_8_2_3_16_1_ENDC_NR_Testbody it is observed that the TTCN configures the SRB3 with SDAP_Configuration_None.

However, as per 38.523-3, SDAP configuration Type is set to OMIT for ENDC testcases. This needs to be corrected

	Summary of change
	SDAP Configuration is set as Omit instead of None. This change is made in addition to the changes specified in R5s201080 for testcase 8.2.3.16.1

	TTCN module
	RRCMeasurement_ENDC_NR

	MCC160 Comment
	


Before Change:

	function fl_TC_8_2_3_16_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig1;
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;
    var PhysCellId v_PhysCellIdServing;
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var template (present) MeasResults v_MeasResults;
    // @siclog "Step 1" siclog@
    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg")); //WA#R5s201080
    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg")); //WA#R5s201080
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_RLC_BearerToAddModList := {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)};
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);
    //WA#R5s201080 : Change Start 
    v_RadioBearerList := {cs_NR_SS_SRB3_Config};
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB3 on NR
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig);  // Now send octetstrings to EUTRA
    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA);
    //WA#R5s201080 : Change End
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);
    //WA#R5s201080 : Change Start 
    // @siclog "Step 3" siclog@
    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3
    v_SCGConfig := cs_NR_CellGroupConfigMeas(tsc_NR_CellGroupId_SCG,  //@sic R5s200612 sic@
                                                   cs_NR_SpCellConfigDef(tsc_NR_CellGroupId_SCG,
                                                                       -,
                                                                       -,
                                                                       cs_NR_ServingCellConfigMeas(tsc_NR_MeasObjectId1)));
    //WA#R5s201080 : Change End
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};
    v_ReportConfigList := cs_NR_ReportConfigToAddModList_1Entry(tsc_NR_IdReportConfigId1, cs_38508_ReportConfigNR_Periodical);
    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config, v_MeasConfig)));
    // @siclog "Step 4" siclog@
    // Check: Does the UE transmits an RRCReconfigurationComplete message via SRB3?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    // @siclog "Step 5" siclog@
    // Check: Does the UE transmits at least 10 MeasurementReport messges to periodically report the measured RSRP value for NR Cell 1
    // via SRB3 within the next 30 seconds?
    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    //WA#R5s201080 : Change Start 
    v_MeasResults := cr_38508_MeasResults(tsc_NR_MeasId1, v_PhysCellIdServing);
    f_NR_ReceivePeriodicalMeasurementReports(nr_Cell1, 30.0, 10, cr_NR_MeasurementReport(v_MeasResults), false);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport_SRB3(nr_Cell1, cr_NR_MeasurementReport(v_MeasResults)));
    //WA#R5s201080 : Change End
    // Test body now finished
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);
  } // End of fl_TC_8_2_3_15_1_TestBody



After Change:

	  function fl_TC_8_2_3_16_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig1;
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModList;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;
    var PhysCellId v_PhysCellIdServing;
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var template (present) MeasResults v_MeasResults;
    // @siclog "Step 1" siclog@
    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertCiphAlg2Bitstring(f_NR_AS_CipheringAlgorithm_Get())), "NR Ciphering Alg")); //WA#R5s201080
    //REMOVED f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (bit2oct(f_NR_ConvertIntAlg2Bitstring(f_NR_AS_IntegrityAlgorithm_Get())), "NR Integrity Alg")); //WA#R5s201080
    v_RadioBearerConfig1 := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, -, -, cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_RLC_BearerToAddModList := {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)};
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, v_RLC_BearerToAddModList);
    //WA#R5s201080 : Change Start 
v_RadioBearerList := {cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB3,cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsSRB),
    











 cs_NR_RlcBearerConfig_SRB1SRB2SRB3(f_NR_LogicalChannelId_SRB(tsc_SRB3), tsc_NR_SRB3_LogicalChannelPriority),omit)};
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_RadioBearerList, cs_TimingInfo_Now); //Configure SRB3 on NR
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig1, v_SCGConfig);  // Now send octetstrings to EUTRA
    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA);
    //WA#R5s201080 : Change End
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);
    //WA#R5s201080 : Change Start 
    // @siclog "Step 3" siclog@
    // The SS transmits an RRCReconfiguration to setup periodical intra-NR measurements and reporting on NR Cell 1 via SRB3
    v_SCGConfig := cs_NR_CellGroupConfigMeas(tsc_NR_CellGroupId_SCG,  //@sic R5s200612 sic@
                                                   cs_NR_SpCellConfigDef(tsc_NR_CellGroupId_SCG,
                                                                       -,
                                                                       -,
                                                                       cs_NR_ServingCellConfigMeas(tsc_NR_MeasObjectId1)));
    //WA#R5s201080 : Change End
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};
    v_ReportConfigList := cs_NR_ReportConfigToAddModList_1Entry(tsc_NR_IdReportConfigId1, cs_38508_ReportConfigNR_Periodical);
    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config, v_MeasConfig)));
    // @siclog "Step 4" siclog@
    // Check: Does the UE transmits an RRCReconfigurationComplete message via SRB3?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    // @siclog "Step 5" siclog@
    // Check: Does the UE transmits at least 10 MeasurementReport messges to periodically report the measured RSRP value for NR Cell 1
    // via SRB3 within the next 30 seconds?
    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    //WA#R5s201080 : Change Start 
    v_MeasResults := cr_38508_MeasResults(tsc_NR_MeasId1, v_PhysCellIdServing);
    f_NR_ReceivePeriodicalMeasurementReports(nr_Cell1, 30.0, 10, cr_NR_MeasurementReport(v_MeasResults), false);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    activate(a_NR_AddDefault_IgnorePeriodicalMeasurmentReport_SRB3(nr_Cell1, cr_NR_MeasurementReport(v_MeasResults)));
    //WA#R5s201080 : Change End
    // Test body now finished
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);
  } // End of fl_TC_8_2_3_15_1_TestBody


	


1.2
 Change 2 : fl_TC_8_2_2_1_1_ENDC_NR_TestBody
	Function name
	fl_TC_8_2_2_1_1_ENDC_NR_TestBody

	Reason for change
	In the current TTCN implementation of fl_TC_8_2_2_1_1_ENDC_NR_Testbody it is observed that the TTCN configures the SRB3 with SDAP_Configuration_None.

However, as per 38.523-3, SDAP configuration Type is set to OMIT for ENDC testcases. This needs to be corrected

	Summary of change
	SDAP Configuration is set as Omit instead of None 

	TTCN module
	RRCReconfig_ENDC_NR

	MCC160 Comment
	


Before Change:

	function fl_TC_8_2_2_1_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) NR_RLC_BearerToReleaseList_Type v_Rlc_BearerToReleaseList := {3};
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var SubFrameTiming_Type v_Timing;
    //@siclog "Steps 1 - 2" siclog@
    //Set the NR ENDC Configuration for SRB3
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, //@sic R5s200419 sic@
                                                         -,
                                                         -,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)});
    //Configure SRB3 in SS
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, {cs_NR_SS_SRB3_Config});//@sic R5s200419 sic@
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig, v_SCGConfig);  // Now send octetstrings to EUTRA
    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA);
    //@siclog "Step 3" siclog@
    //The SS transmits an RRCReconfiguration message to reconfigure NR MAC sent on SRB3/NR Cell 1.
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, -, cs_NR_MAC_CellGroupConfig(-, -, cs_NR_BSR_Config_Step3_822X1));
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    v_Timing := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_NR(v_Timing), cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config)));
    //Reconfigure SS with mac_CellGroupConfig
    f_NR_SS_ServingCellConfig(nr_Cell1, cs_TimingInfo_NR(v_Timing), cs_NR_ServingCellConfig_SpCell(cs_NR_SS_SpCellConfig(omit, omit, v_SCGConfig.mac_CellGroupConfig, omit)));
    //@siclog "Step 4" siclog@
    //Check: Does the UE transmit an RRCReconfigurationComplete message on SRB3/NR Cell 1?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    //@siclog "Steps 5-6" siclog@
    //@sic R5-190465, R5-199030, R5-201230 sic@
    //The SS transmits an RRCReconfiguration message to reconfigure NR PDCP sent on SRB3/NR Cell 1.
    //Check: Does the UE transmit an RRCReconfigurationComplete message on SRB3/NR Cell 1?
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3, cs_38508_PDCP_Config_SRB(ms20))},//@sic R5s200419 sic@
                                                         -,
                                                         -,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig)));
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    // @siclog "Steps 7 - 8" siclog@
    v_RadioBearerConfig := cs_NR_RadioBearer_ReleaseSRB3;
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, v_Rlc_BearerToReleaseList);
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig, v_SCGConfig);  // Now send octetstrings to EUTRA
    //Release SRB3 in SS
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, {cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB3)});
  }


After Change:

	    function fl_TC_8_2_2_1_1_ENDC_NR_TestBody() runs on ENDC_NR_PTC
  {
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) NR_RLC_BearerToReleaseList_Type v_Rlc_BearerToReleaseList := {3};
    var template (omit) octetstring v_EncodedSCG_Config := omit;
    var SubFrameTiming_Type v_Timing;
    //@siclog "Steps 1 - 2" siclog@
    //Set the NR ENDC Configuration for SRB3
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3)}, //@sic R5s200419 sic@
                                                         -,
                                                         -,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, {cs_38508_RLC_BearerConfig_SRB(tsc_SRB3)});
    //Configure SRB3 in SS
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, {cs_NR_SS_RadioBearer_Config(tsc_NR_RbId_SRB3, cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsSRB),
                                                cs_NR_RlcBearerConfig_SRB1SRB2SRB3(f_NR_LogicalChannelId_SRB(tsc_SRB3), tsc_NR_SRB3_LogicalChannelPriority) ,omit)});//@sic R5s200419 sic@
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig, v_SCGConfig);  // Now send octetstrings to EUTRA
    f_ENDC_GetKeysActivateSecurity(nr_Cell1, EUTRA);
    //@siclog "Step 3" siclog@
    //The SS transmits an RRCReconfiguration message to reconfigure NR MAC sent on SRB3/NR Cell 1.
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, -, cs_NR_MAC_CellGroupConfig(-, -, cs_NR_BSR_Config_Step3_822X1));
    v_EncodedSCG_Config := bit2oct(encvalue(v_SCGConfig));
    v_Timing := f_NR_GetNextSendOccasion_DL(nr_Cell1);
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_NR(v_Timing), cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, -, v_EncodedSCG_Config)));
    //Reconfigure SS with mac_CellGroupConfig
    f_NR_SS_ServingCellConfig(nr_Cell1, cs_TimingInfo_NR(v_Timing), cs_NR_ServingCellConfig_SpCell(cs_NR_SS_SpCellConfig(omit, omit, v_SCGConfig.mac_CellGroupConfig, omit)));
    //@siclog "Step 4" siclog@
    //Check: Does the UE transmit an RRCReconfigurationComplete message on SRB3/NR Cell 1?
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    //@siclog "Steps 5-6" siclog@
    //@sic R5-190465, R5-199030, R5-201230 sic@
    //The SS transmits an RRCReconfiguration message to reconfigure NR PDCP sent on SRB3/NR Cell 1.
    //Check: Does the UE transmit an RRCReconfigurationComplete message on SRB3/NR Cell 1?
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB3, cs_38508_PDCP_Config_SRB(ms20))},//@sic R5s200419 sic@
                                                         -,
                                                         -,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary));
    SRB.send(cas_NR_SRB3_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig)));
    SRB.receive(car_NR_SRB3_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    // @siclog "Steps 7 - 8" siclog@
    v_RadioBearerConfig := cs_NR_RadioBearer_ReleaseSRB3;
    v_SCGConfig := cs_38508_CellGroupConfig(tsc_NR_CellGroupId_SCG, -, v_Rlc_BearerToReleaseList);
    f_NR_SendRRCReconfigurationContentsToEUTRA (v_RadioBearerConfig, v_SCGConfig);  // Now send octetstrings to EUTRA
    //Release SRB3 in SS
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, {cs_NR_SS_RadioBearer_Release(tsc_NR_RbId_SRB3)});
  } 

	


